

One-Level PDF Bookmark Created by ODS DOCUMENT and PROC DOCUMENT

Titania Dumas-Roberson, Yanmei Zhang, Saihua Liu, Grifols Inc

ABSTRACT

There are several publications on how to create PDF files with multiple levels of bookmarks. Sometimes it is not convenient for reviewers to review contents with multiple levels of bookmarks. Very often there is a need to create PDF files with only one level of bookmarks which are clear and easy for reviewers to review documents. This paper presents how to meet this need using PROC REPORT without a BY statement or how to create one level of bookmarks using ODS DOCUMENT and PROC DOCUMENT when there is BY statement in PROC REPORT.

INTRODUCTION

Clinical trials have a lot of tables, listings and figures (TLFs). All of them are combined together for review by statistician or other reviewers. Bookmark is quick method for reviewers to find what they want to navigate. However, it is not quick to check TLFs if there are multi-level bookmarks. SAS[®] ODS PDF will automatically create multi-level bookmarks using PROC REPORT. This paper introduces how to create the one level bookmark using PROC REPORT without a BY statement and ODS PDF together and also using ODS DOCUMENT and PROC DOCUMENT when there is BY statement in PROC REPORT.

Creating one-level bookmark using PROC REPORT without by statement

The code below creates the bookmark in the Figure 1 if BOOKMARKGEN =YES is added in ODS PDF statement. There are three nodes in the CONTENTS with the default text in the Figure 1. They are "The Report Procedure", "Detailed and/or summarized report" and "Table 1". Adding ODS PROCLABEL statement controlled the first node. The second node is controlled with the CONTENTS =option on the PROC REPORT statement. The third node can be controlled with the CONTENTS= option on the BREAK BEFORE statement.

```
ods pdf file=tlfout bookmarkgen=Yes style=styles.Custom;
proc report data=final nowd split='!' ls=134 ps=45 missing
style(lines)=[background=white] spanrows;
    column pagbk ord subord colt col ("Study Phase!
_____ " col1 col2 col3 );
    define ord          / order order=data noprint ;
    define subord       / order order=data noprint ;
    define pagbk        / order order=data noprint ;
    define colt         / "Season" order order=internal
style(column)={cellwidth=2in asis =on}
    define col          / "Statistic " style(column)={cellwidth=1.3in}
style(header)={textalign=1 asis = on};
    define col1        / "Run-In! (N=&big1.)" center
style(column)={cellwidth=1.4in} ;
    define col2        / "IV! (N=&big2.)" center
style(column)={cellwidth=1.4in} ;
    define col3        / "SC! (N=&big3.)" center
style(column)={cellwidth=1.4in} ;
    compute after ord  /style={cellheight=0.095in}; line ' ' ; endcomp ;
run ;
```

The code below creates bookmark in the Figure 2. The difference between Figure 1 and Figure 2 code are indicated ⁽¹⁾, ⁽²⁾ and ⁽³⁾ below:

```
ods pdf file=tlfout bookmarkgen=Yes style=Styles.Custom;
ods proclabel "Table 1.1 Summary of Infusions by Season" (1) ;
proc report data=final nowd contents ="Summary"(2) split='!!' ls=134 ps=45
missing
                                style(lines)=[background=white]
spanrows;
  <part of column definition>;
  break before pagbk / contents="My table"(3) page;
  compute after ord /style={cellheight=0.095in}; line '' ;
  endcomp ;
run ;
```




Figure 1. Original Bookmark




Figure 2. Bookmark with the Specific Name

Now we know how to remove the second node and third node from bookmarks. Updating the code above generates one-level bookmark in Figure 3:

```
proc report data=final nowd contents =""(2) split='!!' ls=134 ps=45
break before pagbk / contents=""(3) page;
```




Figure 3. One-Level Bookmark

Creating one-level bookmark using PROC REPORT with BY statement

There are multi-level bookmarks even though (2) and (3) above are set to missing if there is BY statement in the PROC REPORT. There are a lot of unnecessary bookmark generations when there are many categories in one group. For example, the SAS® code below generates Figure 4:

```
filename tlfout "H:\test\bookmark1.pdf" ;
ods pdf file=tlfout bookmarkgen=Yes style=Styles.Custom;
ods proclabel "Table 1.1" ;

title1 j=c "Table 1.1 Summary of Infusions by Season";
title2 j=1 "#byval(agegr1)";
proc report data=final1 nowd contents ="" split='!!' ls=134 ps=45 missing
```

```

style(lines)=[background=white] spanrows;
by agegr1;
  <part of columns>;
break before pagbk / contents="My table" page;
run ;

```




Figure 4. Bookmark with BY statement in the PROC REPORT

SAS® PROC REPORT cannot remove the bookmarks anymore when it has BY statement. We need to rearrange bookmarks by using ODS DOCUMENT and PROC DOCUMENT together to create one level bookmark. Using ODS DOCUMENT is not very different from other ODS features such as ODS PDF or ODS RTF. Saving documents is as simple as putting the ODS DOCUMENT NAME = statement before your code and the ODS DOCUMENT CLOSE statement at the end. The argument that follows the NAME = is the name of the document in which you plan to save your output. There are two advantages of the DOCUMENT procedure. One is replaying output without re-running the original analysis. The other is the ability to manage your output including the BOOKMARK.

The code below assumes that there is not already a document named MYLIB.test_pre. If there is, change the definition of the MYLIB library so that existing work is not overwritten or messed up. The document is not displayed, but saved in a special file called a SAS® item store.

```

ods escapechar="\ ";
ods document name=MYLIB.test_pre(write);
ods pdf notoc file="test1.pdf"; ods listing close;
options nobyline;
ods proclabel "Table 1.1 Summary of Infusions by Season" ;
title1 j=c "Table 1.1 Summary of Infusions by Season";
title2 j=l "#byval(agegr1)";
proc report data=final1 nowd contents ="" split='!' ls=134 ps=45 missing
style(lines)=[background=white] spanrows;
  by agegr1;
  <part of column definitions>;
break before pagbk / contents="" page;
compute after ord/style={cellheight=0.095in}; line '' ; endcomp ;
run ;
ods pdf close;
ods listing;
ods document close;

```

The code below displays a list of all folders and Entries using LIST/levels =all:

```

proc document;
  doc;
  doc name = work.test_pre;
  list/levels=all;

```

```
run;
```

MAKE statement in the PROC DOCUMENT creates one or more new file locations. COPY TO statement copies an entry into specified path. The '^' symbol represents the current path. SETLABEL statement assigns a label to the specified path. Program below shows how to change the LABEL and copy an entry into the path first#1:

```
%macro doc;
  proc document name=work.test_pre(update);
    make \first;
    dir \first#1;
    setlabel \first#1 "Table 1.1 Summary of Infusions by Season";
    %do i=1 %to 2;
      copy \Report#1\ByGroup&i.#1\Report#1 to ^;
    %end;
  run;
%mend doc;
%doc;
```

The REPLAY statement displays one or more entries to the ODS destinations. The Figure 5 is the report with the only one-level bookmark. The code below outputs document in the path first into PDF file:

```
ods pdf close; filename tlfout "H:\test\bookmark1.pdf" ;
ods pdf file=tlfout bookmarkgen=Yes style=Styles.Custom;
proc document name=work.test_pre(update);
  replay \first;
run;
quit;
```

		Study Phase			
Season	Statistic	I	II	III	IV
Total Number of Infusions	n	41	26		
Spring-Summer	n (%)	21 (51.2)	14 (53.8)		
Fall-Winter	n (%)	20 (48.8)	12 (46.2)		

Figure 5. One-Level Bookmark with BY statement in PROC REPORT

CONCLUSION

SAS® can create PDF files with multi-level bookmarks. It is not easy for reviewer to review it. This paper has shown that it is possible to create one level bookmark in PDF using SAS®. Especially, it is hard to generate one level bookmark when using the PROC REPORT with a BY statement. Using ODS DOCUMENT and PROC DOCUMENT within SAS® can rearrange the bookmarks and achieve the goal.

REFERENCES

SAS® 9.2 Output Delivery System User's Guide

<https://support.sas.com/documentation/cdl/en/odsug/61723/PDF/default/odsug.pdf>

Steve Griffiths, GlaxoSmithKline, Stockley Park, "PDF Multi-Level Bookmarks via SAS ", *Proceedings of*

the PharmaSUG 2010 Conference.

<https://www.lexjansen.com/phuse/2010/ts/TS04.pdf>

Tikiri Karunasundera, "Creating and Modifying PDF Bookmarks", *Proceedings of the PharmaSUG 2006 Conference.*

https://www.lexjansen.com/wuss/2006/data_presentation_and_business_intelligence/DPR-Karunasundera.pdf

Ruurd Bennink, OCS Consulting B.V., 's-Hertogenbosch, "ODS DOCUMENT, a practical example", *Proceedings of the PharmaSUG 2015 Conference.*

<https://www.lexjansen.com/phuse/2015/cc/CC01.pdf>

ACKNOWLEDGMENTS

We would like to thank Jane of SAS® Technical Support for their quick and helpful responses.

RECOMMENDED READING

- *Base SAS® Procedures Guide*
- *SAS® For Dummies®*

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Titania Dumas-Roberson
Grifols Inc
79 TW Alexander Dr
Durham, NC 27709
(1) 919-316-6153
Titania.Dumas-Roberson@grifols.com
<http://www.grifolsusa.com/en/web/eeuu#>

Yanmei Zhang
Grifols Inc
79 TW Alexander Dr
Durham, NC 27709
(1) 919-316-2278
Yanmei.zhang@Grifols.com
<http://www.grifolsusa.com/en/web/eeuu#>

Saihua Liu
Grifols Inc
79 TW Alexander Dr
Durham, NC 27709
(1) 919-316-2078
saihua.liu@Grifols.com
<http://www.grifolsusa.com/en/web/eeuu#>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Creating and Customizing Graphics using Graph Template Language, continued

Other brand and product names are trademarks of their respective companies.